

Departamento de Matemática da Universidade de Coimbra		
—/—	Métodos de Programação I	Projecto 0

Cálculo do Instante de Chegada de um Voo

Dado o instante de partida de um avião (horas, minutos, segundos), dado a duração do voo (horas, minutos, segundos), e dado a diferença horária entre a origem e o destino (em horas), calcule o instante de chegada (horas, minutos, segundos), em termos da hora local.

Temos então:

1. Construa um programa em *Pascal* que resolva o problema.
2. Escreva um pequeno relatório (documentação externa) de acordo com o que foi dado nas aulas teóricas.
3. Acrescente ao referido relatório os resultados dos testes que em seu entender são necessários efectuar para uma correcta verificação da correcção do programa.

Notas:

Data de entrega: Data1.

Prazo de realização: Data1+1mês.

Classificação: não conta para a classificação.

Relatório do Projecto 0

Métodos de Programação —/—

Cálculo do Instante de Chegada de um Voo

Trabalho Realizado por: Pedro Quaresma de Almeida.

Especificação: Dado o instante de partida de um avião (horas, minutos, segundos), dado a duração do voo (horas, minutos, segundos), e dado a diferença horária entre a origem e o destino (em horas), calcule o instante de chegada (horas, minutos, segundos), em termos da hora local.

entradas: Instante de partida (hh,mm,ss), Duração do voo (hh,mm,ss), diferença horária entre origem e o destino (hh)

saídas: Instante de chegada (dd, hh, mm, ss)

Utilização: O programa começa por pedir a introdução dos valores respeitantes ao instante de partida, duração do voo, e diferença horária entre a origem e o destino. De seguida o programa calcula e escreve a resposta.

Algoritmo:

```
alg
  ler(hh_part, mm_part, ss_part)
  ler(hh_dur, mm_dura, ss_dur)
  ler(dif_hora)
  somar_horas(hh_part, mm_part, ss_part, hh_dur, mm_dura, ss_dur,
              hh_aux, mm_aux, ss_aux)
  somar_horas(hh_aux, mm_aux, ss_aux, dif_hora, 0, 0,
              hh, mm, ss)
  ajustar(dd, hh)
  se dd = 0 então
    escreve(hh, mm, ss)
  senão
    caso dd seja
      -1 : escrever(hh, mm, ss, ' do dia anterior')
      1 : escrever(hh, mm, ss, ' do dia seguinte')
    fimcaso
  fimse
fimalg
```

O sub-programa `somar_horas` faz a soma de dois instantes de tempo devolvendo o resultado em termos de horas, minutos e segundos, mas sem o ajuste a um dia de 24 horas.

O sub-programa **ajuste** recebe um instante de tempo e faz o ajuste do mesmo para um dia com 24 horas.

Estruturas de Dados: Inteiros.

Particularidades: Não se considera nenhum dos casos em que a diferença entre o instante de partida e o instante de chegada ultrapasse um dia de diferença.

Exemplos: em anexo.

Listagem: em anexo.

Anexo — Exemplos

1. Sem mudança de dia, e no sentido do movimento do sol.

Instante de Partida (hh,mm,ss): 8 23 34
Duracao do Voo (hh,mm,ss): 2 34 16
Diferenca Horaria (hh): 3

A aviao chegou 'as, 13h 57m 50s

2. Sem mudança de dia, e no sentido contrário ao do movimento do sol.

Instante de Partida (hh,mm,ss): 8 23 34
Duracao do Voo (hh,mm,ss): 5 16 0
Diferenca Horaria (hh): -4

A aviao chegou 'as, 9h 39m 34s

3. Com mudança de dia, e no sentido do movimento do sol.

Instante de Partida (hh,mm,ss): 23 34 00
Duracao do Voo (hh,mm,ss): 5 23 16
Diferenca Horaria (hh): 2

A aviao chegou 'as, 6h 57m 16s.
Do dia seguinte.

4. Com mudança de dia, e no sentido contrário ao do movimento do sol.

Instante de Partida (hh,mm,ss): 1 12 45
Duracao do Voo (hh,mm,ss): 1 23 50
Diferenca Horaria (hh): -3

A aviao chegou 'as, 23h 36m 35s.
Do dia anterior.

Anexo — Listagem

```
program voo(input,output);
{
  Programa para calcular o instante de chegada de um dado voo
  entradas: Instante de partida (hhp,mmp,ssp)
            Duracao do voo (hhd,mmd,ssd)
            Diferenca horaria entre origem e o destino (dif_hor)
  saida: Instante de chegada (dd,hhc,mmc,ssc)
}

var
  hhp,mmp,ssp      : integer;
  hhd,mmd,ssd      : integer;
  dif_hor          : integer;
  hhc,mmc,ssc      : integer;
  hh_aux,mm_aux,ss_aux : integer;
  dd               : integer;

procedure somar_horas(  hh1,mm1,ss1,hh2,mm2,ss2 : integer;
                      var hh,mm,ss              : integer);
{
  Procedimento para somar dois instantes de tempo, nao considera
  a mudana de dias.
  entradas: Dois instantes de tempo
  saida: Resultado da soma
}

var
  mm_aux,ss_aux : integer;

begin
  ss_aux := ss1+ss2;
  ss := ss_aux mod 60;
  mm_aux := mm1+mm2+ss_aux div 60;
  mm := mm_aux mod 60;
  hh := hh1+hh2+mm_aux div 60
end; { somar_horas }

procedure ajuste(var dd,hh : integer);
{
  Procedimento para fazer o ajuste referente as mudancas
  de dia.
  entrada: horas (valor eventualmente maior do que 23)
  saidas: dias e horas (entre 0 e 23)
}

begin
  if hh < 0 then
    begin
```

```

        hh := 24+hh;
        dd := -1
    end
else
begin
    dd := hh div 24;
    hh := hh mod 24
end
end; { ajuste }

begin
{ Leitura }
write('Instante de Partida (hh,mm,ss): ');
read(hhp,mmp,ssp);
write('Duracao do Voo (hh,mm,ss): ');
read(hhd,mmd,ssd);
write('Diferenca Horaria (hh): ');
read(dif_hor);
{ Calculo, Instante de partida + duracao }
somar_horas(hhp,mmp,ssp,hhd,mmd,ssd,hh_aux,mm_aux,ss_aux);
{ Calculo (Instante de partida + duracao) + Fusos }
somar_horas(hh_aux,mm_aux,ss_aux,dif_hor,0,0,hhc,mmc,ssc);
{ Ajuste do valor achado, calculo da mudanca de dia }
ajuste(dd,hhc);
{ Escrita do resultado }
writeln;
write('A aviao chegou 'as, ');
if dd = 0 then
    writeln(hhc:2,'h ',mmc:2,'m ',ssc:2,'s')
else
    case dd of
        -1 : begin
            writeln(hhc:2,'h ',mmc:2,'m ',ssc:2,'s','.');
            writeln('Do dia anterior.')
            end;
        1 : begin
            writeln(hhc:2,'h ',mmc:2,'m ',ssc:2,'s','.');
            writeln('Do dia seguinte.')
            end
    end; { case }
end.

```